

Paper 095-2007

## Make Your PROC TABULATE Tables Pretty Using ODS® Style Options

Wendi L. Wright

### ABSTRACT

This intermediate level presentation shows several examples of how to use the ODS Style= option in PROC TABULATE in order to customize tables and improve their attractiveness. This option is very versatile and, depending on where the option is used, can justify cells or row and column headers, change colors for both the foreground and background of the table, modify borders, add a flyover text box in ODS HTML, or add GIF figures to the row or column headers.

### INTRODUCTION

PROC TABULATE is a procedure that displays descriptive statistics in tabular format. It computes many statistics that other procedures also compute, such as MEANS, FREQ, and REPORT and then displays these statistics in a table format. TABULATE will produce tables in up to three dimensions and allows, within each dimension, multiple variables to be reported one after another hierarchically. There are also some very nice mechanisms that can be used to label and format the variables and the statistics produced. This paper does not attempt to teach these options; however, several of the examples will make use of them.

### ODS STYLE ELEMENTS

ODS Style elements can be used to change the look of a table. A few are listed below. These elements all work with HTML, PDF, RTF, PS and PCL destinations. Other elements are listed in the "SAS 9.1 Output Delivery System: User's Guide" that you can find at <http://support.sas.com/documentation/onlinedoc/91pdf/>.

Foreground/Background=	modify color
BorderWidth=	specify thickness of borders
Just/Vjust=	specify horizontal and vertical justification
Font_Face/Font_Weight/Font_Size=	change font characteristics
Rules=	specify horizontal and vertical rule dividers
CellWidth/CellHeight=	change size of table cells
Cellpadding/CellSpacing=	specify white space and thickness of spacing around cell
OutputWidth=	specify width of table

### WHERE DO YOU PUT THE STYLE ELEMENTS?

Depending on where you place the style options, many different results can be achieved. If you place the style options on the PROC TABULATE statement, for example, you will affect all the table cells. See below for a list of some of the different places where you can put the style options and what portion of the table they will affect.

Note that for the CLASS, CLASSLEV, VAR, and KEYWORD statements, the style options can also be specified in the dimension expression in the Table statement.

Style Place In	Part of Table Affected
PROC TABULATE S=[ ...]	data cells
CLASS varname / S=[ ...]	heading for variable varname
CLASSLEV varname / S=[ ...]	class values for variable varname
VAR varname / S=[ ...]	heading for variable varname
KEYWORD stat / S=[ ...]	heading for named stat
TABLE page,row,col / S=[ ...]	table borders, rules, cell spacing
BOX={label=' ' S=[ ...]}	table Box

## CHANGING THE FOREGROUND (TEXT) OR BACKGROUND COLOR

Here is an example showing how to change the color in various parts of the table. Notice that we are applying a black foreground (text) to all cells in the table. For the gender variable we are assigning a background of yellow to the values only (not to the column header) using the CLASSLEV statement. For the mean statistic label we are using a foreground of white and a background of purple.

```
ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black];
  CLASS gender;
  CLASSLEV gender / S=[background=yellow];
  VAR income;
  TABLE gender=' ' all={label='Tot'},
           mean={s=[foreground=white background=purple]} * income
           /      box={label='Income'};
Run;
ODS RTF close;
```

Income	Mean
	Income
Female	52000.69
Male	54089.53
Tot	53011.42

## ADDING JUSTIFICATION

This example shows how to add justification to three sections of the table. We are going to center all the cells inside the table, right justify the total 'Tot' row label, and Bottom Right justify the text inside the upper left box.

You can use ODS Style options to add many things to the table. Here is an example that adds color and justification. We are setting the foreground (print type) to be black and all the cells should be centered. Note the use of the new CLASSLEV statement that we have not seen before.

```
ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black just=c];
  CLASS gender;
  CLASSLEV gender / S=[background=yellow];
  VAR income;
  TABLE gender=' ' all={label='Tot' s=[just=R]},
             mean={s=[foreground=white background=purple]} * income
             /      box={label='Income' s=[VJust=B Just=R]};
Run;
ODS RTF close;
```

	Mean
Income	Income
Female	52000.69
Male	54089.53
Tot	53011.42

## CHANGING CELL WIDTH

This example shows how to change the cell width of both the row headers and the columns of the table.

```
ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black just=c cellwidth=200];
  CLASS gender / S=[cellwidth=250];
  CLASSLEV gender / S=[background=yellow];
  VAR income;
  TABLE gender=' ' all={label='Tot' s=[just=R]},
             mean={s=[foreground=white background=purple]} * income
             /      box={label='Income' s=[VJust=B Just=R]};
Run;
ODS RTF close;
```

	Mean
Income	Income
Female	52000.69
Male	54089.53
Tot	53011.42

## REMOVING THE LINES IN THE TABLE AND ADJUSTING CELL SPACING

With the addition of the rules, cellspacing and cellpadding options we can remove all lines from the table.

```
ODS RTF file='c:\myfile.rtf';
PROC TABULATE data=one f=10.2 S=[foreground=black cellwidth=200 just=c];
  CLASS gender;
  CLASSLEV gender / S=[background=yellow];
  VAR income;
  TABLE gender=' ' all={label='Tot' s=[just=R]},
           /      mean={s=[foreground=white background=purple]} * income
           /      s=[rules=none cellspacing=0 cellpadding=10]
           /      box={label='Income' s=[VJust=B Just=R]};
Run;
ODS RTF close;
```

	Mean
Income	Income
Female	52000.69
Male	54089.53
Tot	53011.42

## ADDING A FLYOVER TEXT BOX USING ODS AND PROC FORMAT

You can use formats with styles to do a number of tricks. One thing you can do is to add a flyover text box to your row or column header cells. The result can't be demonstrated here, but what happens is when you hold your cursor over the column header when you view the table in a Web Browser, a little text box will open underneath that shows the text you have in your format.

```
PROC FORMAT;
  VALUE $ethnic
    'W'='White - Non-Hispanic'
    'H'='Hispanic American'
    'I'='American Indian'
    'A'='Asian'
    'B'='African American'
    ''='Missing'
  ;
ODS HTML file='c:\myfile.HTML';
PROC TABULATE data=one;
  CLASS ethnic gender;
  CLASSLEV ethnic / s=[flyover=$ethnic.];
  VAR income;
  TABLE gender,
         ethnic * income;
RUN;
ODS HTML CLOSE;
```

## USING PROC FORMAT AND ODS TO ADD PICTURES TO YOUR TABLE



Another trick is to add a figure to the row or column header cell. Note that in this example, the gif files needed to be in the same directory that the HTML file was in.

```

PROC FORMAT;
  VALUE $gendGif
    'M'='bluebutterfly.gif'
    'F'='pinkbutterfly.gif';
RUN;

ODS HTML file='c:\myfile.HTML';
PROC TABULATE data=one;
  CLASS GENDER;
  CLASSLEV gender / S=[VJust=T postimage=$gendGif. Cellwidth=80];
  VAR income;
  TABLE income * (N Mean)
    INCOME * MEAN * GENDER;
RUN;
ODS HTML CLOSE;

```

Income		Income	
		Mean	
		Gender	
N	Mean	Female	Male
30	52111.80	 52000.69	 52238.79

## ADDING COLORS TO CELLS BASED ON VALUES IN THE CELLS

The last trick I am going to mention is how to use a format to highlight different cells with different colors based on the final value in the cell.

```

title *Example 30 – Highlighting Cell Values with Colors;
ODS HTML FILE='c:\myfile.html';
PROC FORMAT;
  VALUE watchit
    0 - 20000 = 'Green'
    20001 - 40000 = 'Orange'
    40001 - 50000 = 'Blue'
    50001 - 60000 = 'Purple'
    60001 - high = 'Red' ;
RUN;
PROC TABULATE data=one S=[foreground=watchit.] ;
  CLASS gender ethnic / MISSING;
  VAR income;
  TABLE gender ALL,
    Income * (ethnic ALL) * Mean;
RUN;
ODS HTML CLOSE;

```

	Income						
	Ethnic						All
	Missing	Asian	Af.Amer.	Hisp.	Am. Ind.	White	
	Mean	Mean	Mean	Mean	Mean	Mean	Mean
Gender							
Female	60398.50	26948.00	62083.67	49304.25	56945.00	43321.00	52000.69
Male	57147.50		54039.50	35600.00	78695.00	49334.25	52238.79
All	58773.00	26948.00	58866.00	46563.40	62382.50	47694.27	52111.80

## CONCLUSION

The options described here are by no means exhaustive of what you can do with ODS Style options. Have fun playing with the many options and placement of those options in PROC TABULATE to see how you can make a really fun table.

## AUTHOR CONTACT

Your comments and questions are valued and welcome. Contact the author at:

Wendi L. Wright  
71 Black Pine Ln.  
Levittown, PA 19054-2108  
Phone: (215) 547-3372  
E-mail: [sleepypuppyfeet@earthlink.net](mailto:sleepypuppyfeet@earthlink.net)

SAS and all other SAS Institute Inc. products or service names are registered trademarks or trademarks of SAS Institute, Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.